/*
 * Copyright 2024 the original author or authors.
 *
 *  Licensed under the Apache License, Version 2.0 (the "License");
 *  you may not use this file except in compliance with the License.
 *  You may obtain a copy of the License at
 *      https://www.apache.org/licenses/LICENSE-2.0
 *  Unless required by applicable law or agreed to in writing, software
 *  distributed under the License is distributed on an "AS IS" BASIS,
 *  WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 *  See the License for the specific language governing permissions and
 *  limitations under the License.
 *
 */

package org.spockframework.smoke

import org.codehaus.groovy.ast.ClassNode
import org.jetbrains.annotations.Debug
import org.spockframework.EmbeddedSpecification
import org.spockframework.compiler.InvalidSpecCompileException

class CompileTimeErrorReporting extends EmbeddedSpecification {
  def "constructor declaration"() {
    when:
    compiler.compileSpecBody """
ASpec() {}
    """

    then:
    thrown(InvalidSpecCompileException)
  }

  def "wrong spelling of 'setup' method"() {
    when:
    compiler.compileSpecBody """
def setUp() {}
    """

    then:
    thrown(InvalidSpecCompileException)
  }

  def "'old' method used outside then-block"() {
    when:
    compiler.compileFeatureBody """
when:
def y = old(x)

then:
true
    """

    then:
    thrown(InvalidSpecCompileException)
  }

  def "multiple thrown clauses"() {
    when:
    compiler.compileFeatureBody """
when:
def x = 42

then:
thrown(IllegalArgumentException)
thrown(IOException)
    """

    then:
    thrown(InvalidSpecCompileException)
  }

  def "non-parameterization in where-block"() {
    when:
    compiler.compileFeatureBody """
expect:
true

where:
println "hi"
    """

    then:
    thrown(InvalidSpecCompileException)
  }

  static class Foo {
    Debug getDebug() { return null }
    void bar() {}
  }

  def "compiling with missing dependencies does not fail"() {
    given:
    compiler.addClassImport('org.spockframework.smoke.CompileTimeErrorReporting.Foo')

    when:
    new ClassNode(Class.forName('org.spockframework.smoke.CompileTimeErrorReporting$Foo'))
      .getMethods('bar')

    then: 'make sure it would fail for the right reason, this might change if `Debug` makes it into the runtime classpath'
    NoClassDefFoundError ncdfe = thrown()
    ncdfe.message.contains('org.spockframework.smoke.CompileTimeErrorReporting$Foo')
    ncdfe.message.contains('org/jetbrains/annotations/Debug')

    when:
    compiler.compileFeatureBody """
when:
new Foo().bar()

then:
noExceptionThrown()
    """

    then:
    noExceptionThrown()
  }
}
